

2-Axis Joystick (#27800)

The 2-Axis Joystick can be used to add analog input to your next project. The 2-Axis Joystick contains two independent potentiometers (one per axis) that can be used as dual adjustable voltage dividers, providing 2-Axis analog input in a control stick form. The modular form-factor allows you to plug the 2-Axis Joystick directly into a breadboard for easy prototyping. The 2-Axis Joystick includes spring auto-return to center and a comfortable cup-type knob which gives the feel of a thumb-stick.

Features

- Easy breadboard connection
- Two independent potentiometers with common ground
- Spring auto-return to center position
- Comfortable cup-type knob
- Compatible with most microcontrollers

Key Specifications


- Power capability: 0.01W; 10 VDC maximum working voltage
- Interface: Dual 10 k Ω potentiometers with common ground
- Operating temperature: 32 to 158 °F (0 to 70 °C)
- Dimensions: 1.64" H x 1.40" L x 1.10" W
(41.67 mm H x 35.56 mm L x 27.94 mm W)



Application Ideas

- Camera Pan/Tilt Control
- Game Input/Control
- Robot Control
- Analog Input of Parameters

Quick Start Circuit



This circuit works with the code below for the BASIC Stamp 2 to provide an RCTIME value for each axis that relates to the position of the joystick. In this manner the two potentiometers are providing a variable resistance for use with the RCTIME command. **Caution:** When using this circuit, do not use a resistor value less than 220 Ω and do not apply more than 5 VDC through this resistor to the L/R or U/D pins.

For more information on how to measure resistance using the BASIC Stamp RCTIME command, please read Chapter#5 of *What's a Microcontroller?* book, a free download at www.parallax.com/go/WAM. The PDF is also included in the BASIC Stamp Editor software's Help file, which is a free download from www.parallax.com/basicstampsoftware.

BASIC Stamp[®] 2 Program

```
' {$STAMP BS2}
' {$PBASIC 2.5}

LR VAR Word
UD VAR Word

DO

  HIGH 4
  PAUSE 2
  RCTIME 4, 1, UD


  HIGH 11
  PAUSE 2
  RCTIME 11, 1, LR

  DEBUG HOME, "UD = ", DEC UD, CLREOL, CR,
    "LR = ", DEC LR, CLREOL

  PAUSE 50

LOOP
```

Advanced Circuit



This circuit creates two voltage dividers referenced to VDD (in this case 5 V), using a 2-channel ADC (in this case the MCP3202) to read the voltages at the L/R and U/D pins using the code below. **Caution:** Do not apply voltage to the L/R+ or U/D+ pins that exceeds the I/O pin voltage rating of the device you connect to L/R or U/D, up to 10 VDC maximum. Ground < Analog voltage output at L/R and U/D < VDD.

BASIC Stamp® 2 Program

```
' {$STAMP BS2}
' {$PBASIC 2.5}

CS          PIN      0          ' Chip Select (MCP3202.1)
Clock       PIN      1          ' Clock (MCP3202.7)
DataIn      PIN      2          ' --> Data Out (MCP3202.6)
DataOut     PIN      3          ' --> Data In (MCP3202.5)
Cnts2Mv     CON      $0139      ' x 1.22 (To Millivolts)
result0     VAR      Word       ' Conversion Result CH0
result1     VAR      Word       ' Conversion Result CH1
mVolts0     VAR      Word       ' Result0 --> mVolts
mVolts1     VAR      Word       ' Result1 --> mVolts

DEBUG CLS, "ADC CH 0:", CR, "Volts  :", CR,
          "ADC CH 1:", CR, "Volts  :"

DO
  LOW CS          ' Enable ADC
  SHIFTOUT DataOut, Clock, MSBFIRST, [%1101\4] ' Select CH0, Single-Ended
  SHIFTIN DataIn, Clock, MSBPOST, [result0\12] ' Read ADC
  HIGH CS         ' Disable ADC
  mVolts0 = result0 */ Cnts2Mv ' Convert To Millivolts

  LOW CS          ' Enable ADC
  SHIFTOUT DataOut, Clock, MSBFIRST, [%1111\4] ' Select CH1, Single-Ended
  SHIFTIN DataIn, Clock, MSBPOST, [result1\12] ' Read ADC
  HIGH CS         ' Disable ADC
  mVolts1 = result1 */ Cnts2Mv ' Convert To Millivolts

  DEBUG HOME, CRSRXY, 9, 0, DEC result0, CLREOL,
             CRSRXY, 9, 1, DEC mVolts0 DIG 3,
             ".", DEC3 mVolts0,
             CRSRXY, 9, 2, DEC result1, CLREOL,
             CRSRXY, 9, 3, DEC mVolts1 DIG 3,
             ".", DEC3 mVolts1

  PAUSE 100
LOOP
```